

t_{pw} ein mittlerer Wert (0,3 ms) ergibt; dementsprechend sind mindestens 333 Programmierschleifen nötig. Eine genaue Messung kann man ja leicht mit den Zählern unter Zuhilfenahme einer Uhr durchführen (die den Zählern vorgeschalteten Flipflops teilen durch 4). Bei Handtaktung sorgt der Schmitttrigger S 2 zusammen mit dem 47- μ F-Elektrolytkondensator für eine Entprellung. Der Widerstand begrenzt nur den Entladestrom. Mit dem Rückstellschalter werden alle Flipflops, indirekt auch die Zähler, zurückgesetzt.

Beim Betrieb ergibt sich nun folgender Ablauf:

Mit der Starttaste wird das D-Flipflop FF 1 gesetzt und die Leuchtdiode „Lauf“ an Inverter A 22 leuchtet auf. Falls am Schreib/Lese-Wähler „Schreiben“ eingestellt wurde, wird jetzt über das NAND-Gatter N 6 das Signal WE („Write Enable“) an das EPROM gegeben, die Tristategatter werden geöffnet. In jedem Falle wird die Rückstellung der Zähler aufgehoben. Mit der nächsten positiven Flanke des Taktimpulses wird das D-Flipflop FF 2 gesetzt, wodurch die Zählschaltung aus den IK-Flipflops FF 3, FF 4 freigegeben wird. Diese erzeugen jetzt das in Bild 5a für das Programmieren bzw. Bild 5b für das Prüfen dargestellte Impulsiagramm.

Um eine Programmier- oder Lese-schleife zu beenden, muß FF 1 gelöscht werden. Dies bewirkt die Schlußberkenungslogik aus den Gattern N 5, I 2, I 3, N 1 und N 2, die zusammen ein Nand-gatter bilden, indem sie Null an den D-Eingang von FF 1 legt. Sie ist nötig, weil das Schreibsignal WE schon zurückgenommen werden muß, während noch die letzte programmierte Adresse anliegt. Damit hängt auch die Stufung des Schleifenwählers zusammen:

$$\text{Schleifenzahl} = 2^m + 1$$

mit $m \in \{0, 1, 2, \dots, 11\}$.

(Schluß folgt)

Baudot-Ausgabeprogramm für den 6502

Die beiden am weitesten verbreiteten Codes zur seriellen Datenübertragung sind der Baudot- und ASCII-Code. Unglücklicherweise arbeiten praktisch alle Mikrocomputer im ASCII- und alle Fernschreiber im Baudot-Code; doch stellt dies keinen Grund dar, ein Programm nicht auch einmal von einem preiswert erhältlichen, gebrauchten Fernschreiber ausdrucken zu lassen, wie der folgende Beitrag zeigt.

Die Codes

Der Baudot-Code besteht aus fünf zu übertragenden Datenbits, die das jeweilige Zeichen in serieller Form als Nullen und Einsen enthalten. Das niederwertigste Datenbit wird zuerst, das höchstwertige zuletzt gesendet. Vor

dem ersten Datenbit wird log. 0 als Startbit ausgegeben, nach dem letzten Datenbit folgen 1,5 Stopbits mit log. 1. (1,5 bezieht sich hier auf die Länge; ein Bit dauert bei der üblichen Fernschreibgeschwindigkeit von 50 Baud genau 20 ms.) Ein besonderes Problem

des Baudot-Codes ist, daß die einzelnen Zeichen in sich nicht eindeutig sind; so kann eine bestimmte 5-bit-Folge sowohl einen Buchstaben als auch eine Ziffer oder ein Zeichen darstellen – je nachdem, ob irgendwann vorher einmal ein Zeichen, „Buchstabe“ oder „Ziffer“, gesendet wurde. Der Baudot-Zeichenvorrat ist also zunächst einmal in zwei Teilbereiche getrennt, nämlich die Buchstaben von A bis Z und die Ziffern und Zeichen. Wurde z. B. als letztes Zeichen ein Buchstabe gedruckt, so muß vor der Ausgabe einer Ziffer oder

```

UNTERPROGRAMM ZUR BAUDOT-AUSGABE
0352 86 F5      STX F5      X RETTEN
0354 A2 80      LDX #80    PA 7 =
0356 8E 01 17   STX 1701  AUSGANG
0359 C9 0D      CMP #0D    C.R.
035B D0 04      BNE 0361
035D A9 08      LDA #08
035F D0 0E      BNE 036F
0361 C9 0A      CMP #0A    L.F.
0363 D0 04      BNE 0369
0365 A9 02      LDA #02
0367 D0 06      BNE 036F
0369 29 3F      AND #3F    6 BIT ASCII
036B AA        TAX        ALS INDEX
036C BD C0 03   LDA 03C0, X
036F 85 F4      STA F4
0371 29 20      AND #20    MODUS-TEST
0373 C5 F3      CMP F3     (BU./Z1.)
0375 F0 0E      BEQ 0385
0377 85 F3      STA F3     NEUER MODUS
0379 A8        TAY
037A F0 04      BEQ 0380
037C A9 1B      LDA #1B
037E D0 02      BNE 0382  MODUS-
0380 A9 1F      LDA #1F    UMSCHALTUNG
0382 20 8D 03  JSR 038D  AUSGEBEN
0385 A5 F4      LDA F4     UND ZEICHEN

0387 20 8D 03  JSR 038D  DRUCKEN
038A A6 F5      LDX F5
038C 60        RTS
038D A2 7F      LDX #7F    STARTBIT
038F 8E 00 17   STX 1700
0392 20 B4 03  JSR 03B4
0395 A0 04      LDY #04    5 BITS
0397 4A        LSR
0398 90 04      BCC 039E
039A A2 FF      LDX #FF    '1'
039C D0 02      BNE 03A0
039E A2 7F      LDX #7F    '0'
03A0 8E 00 17   STX 1700
03A3 20 B4 03  JSR 03B4
03A6 88        DEY
03A7 10 EE      BPL 0397
03A9 A2 FF      LDX #FF    STOPBIT
03AB 8E 00 17   STX 1700  (LAENGE =
03AE A2 1E      LDX #1E    1,5 BIT)
03B0 20 B6 03  JSR 03B6
03B3 60        RTS
03B4 A2 14      LDX #14    VERZOEGL. 1 BIT
03B6 8E 07 17   STX 1707
03B9 2C 07 17   BIT 1707
03BC 10 FB      BPL 03B9
03BE 60        RTS
03BF 00        ENDE

```

Bild 1. Unterprogramm zur Ausgabe eines im Akku stehenden ASCII-Zeichens auf einem Baudot-Fernschreiber

TABELLE 6-BIT-ASCII ZU BAUDOT

03C0	00	03	19	0E	09	01	0D	1A
03C8	14	06	0B	0F	12	1C	0C	18
03D0	16	17	0A	05	10	07	1E	13
03D8	1D	15	11	2D	00	3A	00	00
03E0	24	34	00	29	00	24	31	25
03E8	2F	32	39	31	2C	23	3C	3D
03F0	36	37	33	21	2A	30	35	27
03F8	26	38	2E	2B	24	3E	24	39
0400	ENDE							


◀ Bild 2. Tabelle zur Umwandlung eines 6-bit-ASCII- in ein 5-bit-Baudot-Zeichen

ASCII-EINGABE
BAUDOT-AUSGABE

```
0000 20 5A 1E      JSR 1E5A  GETCH(KIM)
0003 20 52 03      JSR 0352  BAUDOT-AUSG.
0006 4C 00 00      JMP 0000  NOCHMAL
```

Bild 3. ▶

Anschluß eines Fernschreiber-Druckers an den Mikrocomputer KIM-1



◀ Bild 4. Einfaches Anwendungsbeispiel: Programm zum Schreiben auf dem TTY-Drucker mit einer ASCII-Tastatur

eines Satzzeichens erst einmal das Zeichen „ZI“ ausgegeben werden.

Ganz anders beim ASCII-Code (American Standard Code for Information Interchange): Jedes Zeichen ist für sich eindeutig definiert; außerdem handelt es sich bei diesem Code normalerweise um ein 7-bit-Format, das $2^7 = 128$ verschiedene Zeichen darstellen kann, wesentlich mehr, als das beim Baudot-Code ($2 \cdot 2^5 = 64$ Zeichen) möglich ist. So sind im ASCII-Code auch Kleinbuchstaben sowie eine ganze Reihe von Steuerzeichen möglich. Aus diesen

Gründen setzte sich ASCII in der Mikrocomputertechnik praktisch hundertprozentig durch.

Das Programm

Aufgabe des in Bild 1 dargestellten Unterprogramms ist es, ein im Akku stehendes ASCII-Zeichen auf einem Baudot-Fernschreiber auszudrucken. Um Platz für die notwendige Umwandlungstabelle (Bild 2) zu sparen, werden nur 6 bit des 7-bit-ASCII-Zeichens als Tabellenindex verwendet; die beiden Steuerzeichen CR (Wagenrücklauf) und LF (neue Zeile) werden vom Programm getrennt decodiert, da sie bei der Einschränkung auf 6 bit verloren gingen. Der 6-bit-Maskierung fallen auch die Kleinbuchstaben zum Opfer; sie können aber auf dem Fernschreiber ohnehin nicht ausgegeben werden.

Das „Haupt-Unterprogramm“ ruft zwei weitere Unterprogramme auf, nämlich eines zur Ausgabe eines im Akku stehenden Baudot-Zeichens (Adresse 038D) und eines zur Vergrößerung um eine der jeweiligen Baud-Rate entsprechende Bit-Dauer, hier für 50 Baud dimensioniert.

Es wurde darauf geachtet, daß die Wirkung dieses Programms auf die Prozessor-Register genau identisch ist mit dem im KIM-Monitor-ROM stehenden Unterprogramm „OUTCH“, d. h. das X-Register wird „gerettet“, Y ist nach der Rückkehr vom Unterprogramm hexadezimal FF, und der Akkuinhalt ist verloren.

Es muß noch erwähnt werden, daß die Tabelle in Bild 2 außer der 5-bit-

Baudot-Information als sechstes Bit noch die Aussage enthält, ob es sich um den Buchstaben- oder um den Ziffernmodus handelt. Ändert sich der Modus gegenüber dem zuletzt ausgegebenen Zeichen, so wird ein entsprechendes Umschaltzeichen an den Fernschreiber ausgegeben. Das Programm benutzt die Zero-Page-Adressen 00F3 (Modus), 00F4 (Baudot-Zeichen) und 00F5 (X-Register).

Notwendige Hardware

Die Timer-Adressen (1707) und die Adressen der I/O-Register beziehen sich auf den Mikrocomputer KIM-1. Das Baudot-Zeichen wird am Anschluß PA 7 ausgegeben; das Programm gestattet es aber, auch einen beliebigen anderen Port als Ausgang zu verwenden. Zum Anschluß des Baudot-Fernschreibers genügt eine Schaltung nach Bild 3. Die Z-Diode dient dazu, Schaltspitzen zu dämpfen, die durch die Selbstinduktion des Empfangsmagneten auftreten. Bild 4 zeigt schließlich ein einfaches Programm, das das direkte Schreiben auf dem Drucker des Fernschreibers mit einem ASCII-Terminal gestattet; es besteht nur aus zwei Unterprogramm-Aufrufen und einem Rücksprungbefehl. Sollte sich herausstellen, daß die Fernschreibergeschwindigkeit nicht 50 Baud beträgt, so sind die hexadezimalen Inhalte der Zellen 03AF und 03B5 zu ändern.

Ein Verlegen des Baudot-Ausgabe-Unterprogramms in eine andere „Page“, also in eine andere Speicherseite, ist sehr einfach: Man braucht nur bei allen 3-Byte-Befehlen, deren drittes Byte 03 lautet, dieses Byte auf die neue Page ändern. Herwig Feichtinger

Programmieren kostet Zeit...

...und Zeit ist Geld. Und deswegen honoriert die FUNKSCHAU jede in der Mikrocomputer-Rubrik veröffentlichte Druckseite mit 200 DM. Dafür erwarten wir aber auch etwas: zum Beispiel einen reproduktionsfähigen Programmausdruck (zur Not genügt auch ein sauber mit der Schreibmaschine abgetipptes Programm) und einen erklärenden Text, was das Programm überhaupt tut. Falls Sie keinen Drucker, aber einen KIM-1 oder einen TI-59 besitzen, genügt es auch, wenn Sie uns Ihr Programm auf Kassette bzw. Magnetkarte übersenden. Und noch eine Möglichkeit: Sie können uns Ihr Programm als Auflistung auch mit dem FSK-Modem auf Kassette überspielen, das in Heft 15/1978 und im Hobbycomputer-Sonderheft des Franzis-Verlages beschrieben war – aber bitte mit 110 Baud und inklusive CR, LF und Leerräumen in ASCII!